
ABSTRACT 
To anticipate trends in retention Kronos’ Hiring Solution divisions utilizes Bureau of Labor Statistics published job 
statistics of hires, separations and job openings available on the internet at http://www.bls.gov/jlt/home.htm or for 
raw data that we utilize at the ftp site;  ftp://ftp.bls.gov/pub/time.series/jt/.   We also utilize the BLS statistic of job 
diffusion for Total Private Non-Farm payrolls available at http://www.bls.gov/webapps/legacy/cesbtab7.htm.   Job 
diffusion is a measure of job creation compared to job destruction.  

Since there are several time series that are related we forecast the BLS time series for hiring, separation and job 
openings using a state space model.  The state space model is appropriate for jointly modeling and forecasting 
several related time series that have dynamic interactions.  By taking into account the autocorrelations among the 
whole set of variables, the state space model may give better forecasts than methods that model each series 
separately. 

INTRODUCTION  
To assist our customers review and anticipate trends in retention we utilize Bureau of Labor Statistics published job 
statistics of hires, separations and job openings available on the internet at http://www.bls.gov/jlt/home.htm or for 
raw data that we utilize at the ftp site;  ftp://ftp.bls.gov/pub/time.series/jt/. We also utilize the BLS statistic of job 
diffusion for Total Private Non-Farm payrolls available at http://www.bls.gov/webapps/legacy/cesbtab7.htm.   Job 
diffusion is a measure of job creation compared to job destruction.  It is indexed from zero to 100 where values 
under 50 represent job destruction and values over 50 represent job creation.   

The Bureau of Labor Statistics (BLS) is the principal fact-finding agency for the Federal Government in the broad 
field of labor economics and statistics. The BLS is an independent national statistical agency that collects, 
processes, analyzes, and disseminates essential statistical data to the American public, the U.S. Congress, other 
Federal agencies, State and local governments, business, and labor. The BLS also serves as a statistical resource 
to the Department of Labor.   

The basic BLS data contains 25 industrial groupings and four types of data; rates and levels.  The ‘level’ refers to 
the count or thousands of hires/separations/job openings.  The ‘rates’ refers to the level plus employed divided by 
employed.  We focus on two industrial groupings from the wealth of data for the four time series of BLS Jobs 
statistics; Total Private (Non-Farm) and Retail Industry.  We focus on these two groupings as most of our clients 
are retail customers and total private non-farm provides overall guidance in job markets and both of these industry 
groupings are seasonally stable and thus provide good forecasts.    

Since there are several time series that are related we forecast the BLS time series for hiring, separation and job 
openings using a state space model.  The state space model is appropriate for jointly modeling and forecasting 
several related time series that have dynamic interactions.  By taking into account the autocorrelations among the 
whole set of variables, the state space model may give better forecasts than methods that model each series 
separately.   

STATE SPACE MODELS 
State space models characterize a multivariate time series through secondary variables, some of which may not be 
observable. The secondary variables are referred to as the state vector and this state vector summarizes the 
present and past values of the time series pertinent to the forecasting. The observed time series are expressed as 
linear combinations of these state variables. State space models are also called a Markovian representation or a 
canonical representation of a multivariate time series process.  

The state space model provides diverse forms of stochastic models. When the dimensions of the predictor space 
are finite, a Gaussian multivariate stationary time series can be written in a state space form. According to Akaike 

(1974) any autoregressive moving average (ARMA) process has a state space representation and any state space 
process can be expressed in an ARMA form.  

SAS PROC STATESPACE® 
SAS® PROC STATESPACE procedure is based on Akaike (1976). The form of the state space model used by the. 
The model is defined by the following state transition formulation:  

 zt+1 = F * zt + G * (et+1) 

The matrix F is called the transition matrix and is referred to as the state transition equation with size s × s 
coefficients and it determines the dynamic nature of the model.  The matrix G is called the input matrix is a matrix 
of  s × r coefficient and determines the variance structure of the transition equation. In model identification, the first 
r rows and columns of G are set to an r × r identity matrix.  The random error input vector et, sometimes called the 
innovation vector or shock vector, is a sequence of independent normally distributed random vectors of dimension r 
with mean 0 and covariance matrix. 

State space models usually include a measurement or observation equation giving the observed values xt as a 
function of the state vector zt. In SAS ® PROC STATESPACE the measurement equation is merely an extraction of 
the first r components of the state vector and is always includes the observed values xt in the state vector zt.  

In summary:  

xt is an observation vector of dimension r.  

zt is a state vector of dimension s, whose first r elements are xt and whose last s-r elements are 
conditional prediction of future xt.  

  where xt = [ Ir 0 ] zt 

F is the s × s transition matrix.  

G is the s × r input matrix with the Ir (identity matrix) forming the first r rows and columns.  

et are independent normally distributed random vectors of dimension r with mean 0 and a covariance 
matrix.  

HOW SAS PROC STATESPACE® WORKS 
The SAS ® PROC STATESPACE procedure is designed following the modeling strategy proposed by Akaike 
(1976). The strategy employed considers the canonical correlation analysis for the automatic identification of the 
state space model.  

Akaike (1976) recommends first fitting a sequence of unrestricted vector autoregressive models and computes 
Akaike's information criterion (AIC) for each model.  Using sample autocovariance matrices and the Yule-Walker 
equations to model the vector autoregressive models and the order of the VAR model producing the smallest 
Akaike information criterion is chosen as the order (number of lags into the past) to use in the canonical correlation 
analysis.  

Through the selected order the elements of the state vector selected via a series of canonical correlation analyses 
of the observed autocovariance matrices. This analysis computes the sample canonical correlations of the past with 
an increasing number of steps into the future. Variables that yield insignificant correlations are excluded from 
further consideration and those that yield significant correlations are added to the state vector. Another information 
criterion proposed by Akaike is used to evaluate the importance of the correlation. When specifying the state vector 
explicitly the model identification steps are omitted.  

After identification of the state vector, the state space model is fit to the data. The estimated of the parameters is 
done by approximate maximum likelihood for the free parameters in the F, G and covariance matrix. By default, and 
except for identifiability requirements, the F and G matrices are unrestricted,. And, optionally, conditional least-
squares estimates can be computed and restrictions imposed on elements of the F and G matrices.  

Using the Kalman filtering technique forecasts are produced after the parameters are estimated and from the fitted 
state space model.  In the presence of differencing the forecasts are integrated to produce forecasts from the 
original input variables. 

ANALYSIS OF BLS JOB STATISTICS 
After downloading the BLS series data from the FTP website, we use SAS to load and combine the data for state 
space modeling.  The only filtering we do is to limit our forecasting to the data that is not seasonally adjusted and to 
select and model only the Total Private and Retail industrial groupings of the BLS data.  Our forecasting approach 
consists of the following: 

1) Forecast the rates of hires, separations and job openings for each grouping separately 

2) Forecast the levels of hires, separations and job openings for each grouping separately 

3) Forecast the diffusion levels considering the hires, separations and job openings for Total Private industry 
grouping.  The only diffusion level from BLS of interest to us is Total Private Non-Farm.  

For PROC STATESPACE we need to create stationary time series.  This is easily done with the 2 industrial 
groupings; both show significant annual seasonality that can be remedied with a lag of 12 and these industrial 
groupings show ongoing month over month growth which is remedied using a lag of 1.  This differencing is 
accomplished within the STATESPACE procedure with the “(1,12)” specification following the variables included in 
the model.  Retail and total private industries follow strong seasonality and therefore a lag of 12 for the annual 
seasonality.  We output the data for use in graphing the subsequent forecasts and in the analysis we request a 12 
month forecast from the modeling.    

PROC STATESPACE automatically selects the appropriate models, fits the differenced data for F and G described 
above for forecasting, and forecasts as requested the months into the future.  The output from SAS’s PROC 
STATESPACE is rather voluminous and can be provided upon request.  Here we will show only the transition matrix 
F and the interpretation along with graphical forecasts for Job Statistics. 

SAS PROC STATESPACE® SET UP FOR BLS DATA 
SAS PROC STATESPACE® provides (as all SAS® procedures) many SAS PROC STATESPACE procedure 
options.  The options fall into the following categories; printing, input data, preliminary autoregressive models, 
canonical correlations analysis, state space estimation and forecasting options. Given the monthly definition of the 
BLS data series, we found the following parameter set up to be intuitive and natural implementation.  Interpretation 
of the parameter settings is given below the SAS ® code. 

proc statespace  
 data=work.subset_jolt_x  out=work.pred  
 lead=12      interval=month  
 LAGMAX=12  ARMAX=12  NOCENTER  DIMMAX=12  cancorr;  
by industry_text; 
label hire_level="Hire Level"  
    totsep_level="Separation Level"  
    jobop_level="Job Open Level"; 
 var hire_level(1,12) totsep_level(1,12) jobop_level(1,12); 
  id month_id; 
 run; 
 

Interpretation of SAS PROC STATESPACE® of the parameter settings 

a) data=work.subset_jolt_x  the input dataset 
b) out=work.pred   output dataset for forecast 
c) lead=12    number of months forecasted into the future.  

(default is 0 or none) This is a Forecast 
setting. 

d) interval=month   Specifies the time interval between 
observations. Used in conjunction with the ID 
variable to check that the input data are in 
order and have no missing periods. This is a 
Forecast setting. 

e) LAGMAX=12   Input dataset parameter to calculate the 
sample autocovariance.  BLS data is highly 
seasonal so the 12 months lag is natural. (The 

default is 10) This is an Input parameter 
setting. 

f) ARMAX=12   part of the preliminary autoregressive model 
this is used to select autoregressive lags to be 
evaluated and selected for forecasting. (The 
default is 10)  This is a Preliminary 
Autoregressive parameter. An ARMAX=6 is used in 
our SAS PROC STATESPACE implementation for the 
forecasting of the BLS Diffusion index. 

g) NOCENTER   Works with LAGMAX and prevents subtraction of 
the sample mean from the input series (after any 
specified differencing).  This is an Input 
parameter setting 

h) DIMMAX=12   Upper dimension of the state vector in the 
canonical correlation analysis.  (Again the 
default is 10) 

i) CANCORR  Prints the canonical correlations and 
information criteria for the candidate state 
vectors considered. 

j) PASTMIN=6    Specifies the number of minimum lags to 
include in the canonical correlations.  Used in 
our SAS PROC STATESPACE implementation for the 
forecasting of the BLS Diffusion index.   

 

JOLT (JOB OPENINGS AND LABOR TURNOVER) DATA LOADING 
Load the JOLT (job openings and labor turnover) data series and relevant data dimensions is easy and fast using 
SAS® PROC IMPORT after setting up a library for the loaded data.  But in order to use the data some of the loaded 
data variable names are renamed in the SAS® PROC DATASETS.  After merging in each of the relevant data 
dimension sources of data element types, rate level codes region and seasonality codes a level (how many) and a 
rates (relatively how much) data set are created for each type of the three series; hiring, total separations and job 
openings. 

libname SAS_jolt "S:\stratall\Analytics Research Projects\SAS Libraries"; 
 
%let JT_industry="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.industry"; 
%let JT_dataelement="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.dataelement"; 
%let JT_ratelevel="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.ratelevel"; 
%let JT_region="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.region"; 
%let JT_seasonal="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.seasonal"; 
%let JT_footnote="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.footnote"; 
%let JT_current="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.data.0.Current"; 
%let JT_all="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.data.1.AllItems"; 
%let JT_series="S:\stratall\Analytics Research Projects\Bureau of Labor 
Statistics BLS\JT Files\jt.series"; 
 
 
 
PROC IMPORT OUT= SAS_jolt.JT_industry  

DATAFILE= &JT_industry DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  
DATAROW=2; RUN; 
 
PROC IMPORT OUT= SAS_jolt.JT_dataelement  
DATAFILE= &JT_dataelement DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  
DATAROW=2; RUN; 
 
PROC IMPORT OUT= SAS_jolt.JT_ratelevel  
DATAFILE= &JT_ratelevel DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  
DATAROW=2; RUN; 
 
PROC IMPORT OUT= SAS_jolt.JT_region 
DATAFILE= &JT_region DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  
DATAROW=2; RUN; 
 
PROC IMPORT OUT= SAS_jolt.JT_seasonal 
DATAFILE= &JT_seasonal DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  
DATAROW=2; RUN; 
 
PROC IMPORT OUT= SAS_jolt.JT_footnote  
DATAFILE= &JT_footnote DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  
DATAROW=2; RUN; 
 
PROC IMPORT OUT= SAS_jolt.JT_current  
DATAFILE= &JT_current DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  
DATAROW=2; RUN; 
 
PROC IMPORT OUT= SAS_jolt.JT_all 
DATAFILE= &JT_all DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  DATAROW=2; 
RUN; 
 
PROC IMPORT OUT= SAS_jolt.JT_series 
DATAFILE= &JT_series DBMS=DLM REPLACE; DELIMITER='09'x; GETNAMES=YES;  
DATAROW=2; RUN; 
 
proc datasets library=sas_jolt;  
  modify jt_footnote; 
  rename  footnote_code=footnote_codes;  
  modify jt_dataelement; 
   rename  display_level= data_display_level 
   selectable= data_selectable 
   sort_sequence= data_sort_sequence;  
  modify jt_industry; 
   rename  display_level= ind_display_level 
   selectable= ind_selectable 
   sort_sequence= ind_sort_sequence;  
  modify jt_ratelevel; 
   rename  display_level= rate_display_level 
   selectable= rate_selectable 
   sort_sequence= rate_sort_sequence;  
  modify jt_region; 
   rename  display_level= regn_display_level 
   selectable= regn_selectable 
   sort_sequence= regn_sort_sequence;  
run;  
quit; 

proc sort data=sas_jolt.jt_series; by industry_code;run; 
proc sort data=sas_jolt.jt_industry; by industry_code;run; 
 
data sas_jolt.jt_series2; 
merge sas_jolt.jt_series (in=inseries) sas_jolt.jt_industry;  
by industry_code; if inseries; 
run; 
 
proc datasets library=sas_jolt; delete jt_series; run; quit; 
 
proc sort data=sas_jolt.jt_series2; by region_code;run; 
proc sort data=sas_jolt.jt_region; by region_code;run; 
 
data sas_jolt.jt_series3; 
merge sas_jolt.jt_series2 (in=inseries) sas_jolt.jt_region;  
by region_code; if inseries; 
run; 
 
proc datasets library=sas_jolt; delete jt_series2; run; quit; 
 
proc sort data=sas_jolt.jt_series3; by dataelement_code;run; 
proc sort data=sas_jolt.jt_dataelement; by dataelement_code;run; 
 
data sas_jolt.jt_series4; 
merge sas_jolt.jt_series3 (in=inseries) sas_jolt.jt_dataelement;  
by dataelement_code; if inseries; 
run; 
 
proc datasets library=sas_jolt; delete jt_series3; run; quit; 
 
proc sort data=sas_jolt.jt_series4; by ratelevel_code;run; 
proc sort data=sas_jolt.jt_ratelevel; by ratelevel_code;run; 
 
data sas_jolt.jt_series5; 
merge sas_jolt.jt_series4 (in=inseries) sas_jolt.jt_ratelevel;  
by ratelevel_code; if inseries; 
run; 
 
proc datasets library=sas_jolt; delete jt_series4; run; quit; 
 
proc sort data=sas_jolt.jt_series5; by footnote_codes;run; 
proc sort data=sas_jolt.jt_footnote; by footnote_codes;run; 
 
data sas_jolt.jt_series_all; 
merge sas_jolt.jt_series5 (in=inseries) sas_jolt.jt_footnote;  
by footnote_codes;  
if inseries; 
rename footnote_text  = series_footnote_text ; 
rename footnote_codes = series_footnote_code; 
run; 
 
proc datasets library=sas_jolt; delete jt_series5; run; quit; 
 
proc sort data=sas_jolt.jt_all; by footnote_codes;run; 
proc sort data=sas_jolt.jt_footnote; by footnote_codes; run; 
 
data sas_jolt.jt_all1; 

merge sas_jolt.jt_all (in=inseries) sas_jolt.jt_footnote; 
by footnote_codes; 
if inseries; 
rename footnote_text  = all_footnote_text; 
rename footnote_codes = all_footnote_code; 
run; 
 
proc sort data=sas_jolt.jt_all1; by series_id year period;run; 
proc sort data=sas_jolt.jt_series_all; by series_id; run; 
 
data sas_jolt.jt_all_x; 
merge sas_jolt.jt_all1 (in=inall) sas_jolt.jt_series_all; 
by series_id; 
if inall; 
retain series_seq; 
if first.series_id then series_seq=1; else series_seq=series_seq+1; 
run; 
 
proc datasets library=sas_jolt; delete jt_all1; run; quit; 
 
data work.jt_hire_level  (rename=(value=hire_level)) 
  work.jt_hire_rate   (rename=(value=hire_rate)) 
   work.jt_totsep_level (rename=(value=totsep_level)) 
    work.jt_totsep_rate (rename=(value=totsep_rate)) 
     work.jt_jobop_level (rename=(value=jobop_level)) 
   work.jt_jobop_rate (rename=(value=jobop_rate)); 
set sas_jolt.jt_all_x; 
keep industry_code industry_text month_id value; 
period_id= substr(period,2,2)*1; 
dayof=1; 
if period_id>12 then delete; 
Month_id = mdy(period_id,dayof,year); 
drop period_id dayof; 
 
if seasonal eq "U"; 
if industry_code in (100000,110099,230000,300000,320000,340000, 
 400000,420000,440000,480099,510000,510099,520000,530000,540099,   
 600000,610000,620000,700000,710000,720000,810000,900000,910000,920000); 
if dataelement_code in ("HI","TS","JO"); 
 
if dataelement_code eq "HI" then 
 if ratelevel_code eq "L" then output work.jt_hire_level; 
  else output work.jt_hire_rate; 
 ELSE 
if dataelement_code eq "TS" then 
 if ratelevel_code eq "L" then output work.jt_totsep_level; 
  else output work.jt_totsep_rate; 
 ELSE 
if dataelement_code eq "JO" then 
 if ratelevel_code eq "L" then output work.jt_JOBOP_level; 
  else output work.jt_JOBOP_rate; 
 ELSE DELETE; 
RUN; 
 

DIFFUSION INDEX DATA LOADING 
Loading the BLS Diffusion Index data series is a little trickier.  Using the data provided through BLS’s data 
extraction process an MS Excel® spreadsheet of history is maintained and then saved to a comma separated file 

for easy reading into SAS.   Occasionally the BLS will update the methodology used to create the series and they 
will ‘recast’ or recreate the history.  The data will be resaved as needed and MS Excel® provides a simple interface 
for maintaining the database of the BLS Diffusion Index. 

 
 
PROC IMPORT OUT= WORK.JT_diff  
 DATAFILE= "S:\stratall\Analytics Research Projects\Bureau of Labor Statistics 
BLS\DiffusionINDEXIN.csv" DBMS=CSV REPLACE; GETNAMES=YES; DATAROW=2; RUN; 
 
 
data work.JT_diff; 
 SET jt_diff; 
 length industry_code 8.; 
 length industry_text $51.; 
 industry_code = 100000; 
 industry_text = "Total Private"; 
 month_id = mdy(monthnum,1,yearnum); 
 drop monthnum period yearnum series_id; 
 run; 
 
proc sort data=work.jt_hire_level; by industry_code month_id; run; 
proc sort data=work.jt_hire_rate; by industry_code month_id; run; 
proc sort data=work.jt_totsep_level; by industry_code month_id; run; 
proc sort data=work.jt_totsep_rate; by industry_code month_id; run; 
proc sort data=work.jt_jobop_level; by industry_code month_id; run; 
proc sort data=work.jt_jobop_rate; by industry_code month_id; run; 
proc sort data=work.jt_diff; by industry_code month_id; run; 
 
data sas_jolt.jt_all2 ; 
merge  work.jt_hire_level  work.jt_hire_rate   
   work.jt_totsep_level    work.jt_totsep_rate 
     work.jt_jobop_level   work.jt_jobop_rate work.jt_diff; 
by industry_code month_id; 
label hire_rate="Hire Rate"; 
label totsep_rate="Separation Rate"; 
label jobop_rate="Job Opening Rate"; 
label hire_level="Hire Level"; 
label totsep_level="Separation Level"; 
label jobop_level="Job Opening Level"; 
label diff_12mth="12 Month Diffusion"; 
label industry_text="Industry"; 
hire_rate_lag1 = lag(hire_rate); 
hire_level_lag1 = lag(hire_level); 
run; 

 
 

SAS® CODE FOR SAS PROC STATESPACE® AND BLS DATA  
Once the data is squished together into one data set we can begin the use of SAS PROC STATESPACE®.  The 
SAS® code uses the steps below uses the following logic: 

1) Using macro variables set the minimum and maximum dates for plotting (yes, we use SAS PROC 
GPLOT®).  These have to be changed relative to each month the process is run and the BLS data has 
been updated. 

2) Create a dataset with BLS Total Private and Retail Industry data only. 

3) Using SAS ODS ® open two files, one that’s an rich test (RTF) and one that a PDF  

a) In RTF’s the SAS ODS ® output can be easily copied/pasted 

b) We find PDFs handle SAS ODS® graphs better than RTFs 

4) Execute SAS PROC STATESPACE® on JOLT Levels (number of…) data 

5) Plot JOLT history and forecast from SAS PROC STATESPACE® 

6) Execute SAS PROC STATESPACE® on JOLT Rates (relative number of… ) data 

7) Plot JOLT history and forecast from SAS PROC STATESPACE® 

8) Execute SAS PROC STATESPACE® on JOLT Rates and 12 Month Diffusion Index data 

9) Plot 12 Months Diffusion Index history and forecast from SAS PROC STATESPACE® 

10) At the end make sure to close the rtf and pdf files while re-opening the listing output. 

 
data _null_; 
 call symput('keydate','01Jun2009'd); 
 call symput('mindate','01Jun2004'd); 
 call symput('minDIFdate','01Jun2001'd); 
 call symput('maxdate','01Jun2009'd); 
run; 
 
proc sort data=sas_jolt.jt_all2;  by industry_text month_id; run; 
data work.subset_jolt_x; 
 set sas_jolt.jt_all2; 
 where industry_code in (100000, 440000); 
  run; 
 
ods listing close; 
ods rtf file='\\file1\mjetton$\My SAS Files\9.1\BLS_statespace1.rtf'; 
ods pdf file='\\file1\mjetton$\My SAS Files\9.1\BLS_Graphs.pdf'; 
 
title 'Forecast of BLS Stats' height=1; 
 
proc statespace data=work.subset_jolt_x  
 out=work.pred lead=12   
 interval=month  
 LAGMAX=12  
 ARMAX=12  
 NOCENTER  
 DIMMAX=12  
 cancorr;  
 by industry_text; 
 label hire_level="Hire Level"  
    totsep_level="Separation Level"  
    jobop_level="Job Open Level"; 
 var hire_level(1,12) totsep_level(1,12) jobop_level(1,12); 
  id month_id; 
 run; 
 
symbol1 i=needle; 
symbol2 i=needle; 
symbol3 i=needle; 
axis1 label=(angle=0 h=1 color=black 'Month') minor=NONE; 
axis2 label=(angle=90 h=1 color=black 'Level (Thousands)'); 
 
data work.pred_scrubbed; 
set work.pred; 
if month_id lt &keydate then 

 do; for1=.;for2=.;for3=.; end; 
label for1="Forecast of Hire Level"; 
label for2="Forecast of Separation Level"; 
label for3="Forecast of Job Opening Level"; 
run; 
 
proc gplot data=work.pred_scrubbed; 
title2 'Forecast of Next 12 Months' height=.1; 
title3  'LEVELS' height=.1; 
by industry_text; 
where month_id > &mindate; 
format month_id MONYY5.; 
format hire_level comma6.0; 
format totsep_level comma6.0; 
format jobop_level comma6.0; 
symbol1 color=green interpol=spline 
        width=0.5 value=triangle 
        height=0.5; 
symbol2 color=purple interpol=spline line=2 
        width=3 value=circle 
        height=0.5; 
symbol3 color=red interpol=spline 
        width=0.5 value=square 
        height=0.5; 
symbol4 color=magenta interpol=spline line=2 
        width=3 value=square 
        height=0.5; 
symbol5 color=black interpol=spline 
        width=0.5 value=square 
        height=0.5; 
symbol6 color=yellow interpol=spline line=2 
        width=3 value=square 
        height=0.5; 
legend1 label=(position=bottom justify=center); 
 plot hire_level * month_id  
 for1 * month_id  
 totsep_level * month_id  
 for2 * month_id  
 jobop_level * month_id  
 for3 * month_id  
   /grid overlay legend=legend1 haxis=axis1 vaxis=axis2; 
 run; 
 quit; 
 
title 'Forecast of BLS Statistics'; 
 
proc statespace data=work.subset_jolt_x  
 out=work.pred_rates  
 lead=12  
 interval=month  
 LAGMAX=12  
 ARMAX=12  
 NOCENTER  
 DIMMAX=12   
 cancorr;  
 by industry_text; 

label hire_rate="Hire Rate" totsep_rate="Separation Rate" jobop_rate="Job Open 
Rate"; 
var hire_rate(1,12) totsep_rate(1,12) jobop_rate(1,12); 
id month_id; 
run; 
 
axis2 label=(angle=90 h=1 color=black 'Rate'); 
 
data work.pred_scrubbed_rates; 
set work.pred_rates; 
if month_id lt &keydate then 
 do; for1=.;for2=.;for3=.; end; 
label for1="Forecast of Hire Rate"; 
label for2="Forecast of Separation Rate"; 
label for3="Forecast of Job Opening Rate"; 
run; 
 
proc gplot data=work.pred_rates; 
title2 'Forecast 12 Months' height=.1; 
title3 'RATES' height=.1; 
 by industry_text; 
where month_id > &mindate; 
format month_id monyy5.; 
format hire_rate comma3.1; 
format totsep_rate comma3.1; 
format jobop_rate comma3.1; 
label for1="Forecast of Hire Rate"; 
label for2="Forecast of Separation Rate"; 
label for3="Forecast of Job Opening Rate"; 
symbol1 color=green interpol=spline 
        width=0.5 value=triangle 
        height=0.5; 
symbol2 color=green interpol=spline line=2 
        width=1 value=circle 
        height=0.5; 
symbol3 color=red interpol=spline 
        width=0.5 value=square 
        height=0.5; 
symbol4 color=red interpol=spline line=2 
        width=1 value=square 
        height=0.5; 
symbol5 color=black interpol=spline 
        width=0.5 value=square 
        height=0.5; 
symbol6 color=black interpol=spline line=2 
        width=1 value=square 
        height=0.5; 
legend1 label=(position=bottom justify=center); 
 plot hire_rate * month_id  
 for1 * month_id  
 totsep_rate * month_id  
 for2 * month_id  
    jobop_rate * month_id  
 for3 * month_id  
   /grid overlay legend=legend1 haxis=axis1 vaxis=axis2; 
 run; 
 quit; 

data work.subset_jolt_xx; 
 set sas_jolt.jt_all2; 
 where industry_code = 100000; 
 run; 
 
proc statespace data=work.subset_jolt_xx  
 out=work.pred_diff  
 lead=12  
 interval=month  
 LAGMAX=12  
 ARMAX=6  
 NOCENTER  
 DIMMAX=12 
 pastmin=6   
 cancorr;  
by industry_text; 
label hire_rate="Hire Rate"  
 totsep_rate="Separation Rate"  
 jobop_rate="Job Open Rate"  
 diff_12mth="Diffusion 12 Mth"; 
var hire_rate(1,12) totsep_rate(1,12) jobop_rate(1,12) diff_12mth(1,12) ; 
 id month_id; 
 run; 
 
axis1 label=(angle=0 h=1 color=black 'Month') minor=NONE ; 
axis2 order=(0 to 100 by 10) label=(angle=90 h=1 color=black 'Diffusion'); 
 
data work.pred_scrubbed_diff; 
set work.pred_diff; 
if month_id lt &keydate then 
 do; for1=.;for2=.;for3=.;for4=.; end; 
 upp_for4 = for4+2*std4; 
 lwr_for4 = for4-2*std4; 
 label for4="Forecast of 12 Mth Diffusion"; 
 label upp_for4="+2 STD of Forecast"; 
 label lwr_for4="-2 STD of Forecast"; 
run; 
 
proc gplot data=work.pred_scrubbed_diff; 
title2 'Forecast 12 Months' height=.1; 
title3 'DIFFUSION' height=.1; 
 by industry_text; 
where month_id > &mindifdate; 
format month_id monyy5.; 
symbol1 color=green interpol=spline 
        width=1 value=triangle 
        height=0.5; 
symbol2 color=purple interpol=spline line=2 
        width=1 value=circle 
        height=0.5; 
legend1 label=(position=bottom justify=center); 
 plot diff_12mth * month_id  
 for4 * month_id  
  /grid overlay legend=legend1 haxis=axis1 vaxis=axis2; 
 run; 
 quit; 

ods rtf close; 
ods pdf close; 
ods listing; 
 

CONCLUSIONS 
The Bureau of Labor Statistics time series data with the seasonality and trending are a natural pairing for playing 
with SAS PROC STATESPACE® for the interpretation of the retail job and total job market.  The easy access 
provided by the BLS to a wealth of data and the SAS® tool set provide a great learning environment for 
practitioners and educators. 

BLS Retail Industry Hiring and Separation Rates and Levels forecasting depends only upon prior period value after 
being de-trended for a 1 month and 12 month lags.  Retail Job Opening Rate and Level models depend upon prior 
period value and the error in the prediction of prior period Job Openings.   

BLS Total Private Industry Hiring, Separation and Job Opening Rates and Levels forecasting depend upon prior 
period value after being de-trended for a 1 month and 12 month lags.   Job Openings are an indicator of Hiring 
Rates and Levels in the state space models. 

